Code for RGD-^{125}IPt-AuNRs

```
STARTTIME= 0
STOPTIME=30
DTMAX = 0.1
DTOUT =0.1

; Physiological constants (kg)
BW   = 0.02                    ; body weight

; Dose (% ID)
IV = 0.1

; Organ volumes (fraction of body weight)
VLuC=0.007                     ; Lung; Brown et al, 1997[1]
VLC=0.0549                     ; Liver; Brown et al, 1997[1]
VSC=0.005                      ; Spleen; Davies and Morris 1993[2]
VKC=0.0167                     ; Kidneys; Brown et al. 1997[1]
VTC=0.015                      ; Tumor; Baxter et al. 1994[3]
VBloodC=0.085                  ; Blood; Davies and Morris 1993[2]
VBoC=1-(VLC+VKC+VSC+ VTC+VBloodC)    ; Rest of body

; Organ volumes (L)
VBlood=VBloodC*BW
VLu=VLuC*BW                    ; Lung
VL=VLC*BW                      ; Liver
VS=VSC*BW                      ; Spleen
VK=VKC*BW                      ; Kidneys
VT=VTC*BW                      ; Tumor
VA = 0.2*VBloodC*BW            ; Arterial blood
VV = 0.8*VBloodC*BW            ; Venous blood
VBo=VBoC*BW                    ; Rest of body
```

```
; Organ blood volumes (L)
VLuVES = 0.5*Vlu                    ; Lung; Brown et al, 1997
VLVES = 0.31*VL                     ; Liver; Brown et al, 1997
VSVES = 0.17*VS                     ; Spleen; Brown et al, 1997
VKVES = 0.24*VK                     ; Kidneys; Brown et al, 1997
VTVES = 0.07*VT                     ; Tumor; Baxter et al. 1994
VBoVES = 0.04*VBo                   ; Rest of body; Lin et al, 2015; Brown et al, 1997

; Organ tissue volumes (L)
VLuT = 0.5*Vlu                      ; Lung; Calculated from Davda et al, 2008
VLT = 0.69*VL                       ; Liver; Calculated from Davda et al, 2008
VST = 0.83*VS                       ; Spleen; Calculated from Davda et al, 2008
VKT = 0.76*VK                       ; Kidneys; Calculated from Davda et al, 2008
VTT = 0.93*VT                       ; Tumor; Baxter et al. 1994
VBoT = 0.96*VBo                     ; Rest of body; Calculated from Davda et al, 2008


; Blood flow rate (fraction of cardiac output)
QCC=16.5                    ; Cardiac output constant; Brown et al, 1997
QLC=0.161                   ; Liver; Brown et al, 1997.
QKC=0.091                   ; Kidney; Brown et al, 1997
QSC=0.01125                 ; Spleen; Lin et al, 2015; Davies and Morris, 1993
QTC=0.0225                  ; Tumor; Brown et al, 1997
QBoC=1-(QLC+QSC+QKC+ QTC) ; Rest of body

; Cardiac output and regional blood flow (L/h)
QC=QCC*BW^0.75
QLu=QC                      ; Lung
QL=QLC*QC                   ; Liver
QS=QSC*QC                   ; Spleen
QK=QKC*QC                   ; Kidneys
```

QT=QTC*QC ; Tumor

QBo=QBoC*QC ; Rest of body

; Distribution coefficients (PC), unitless

PLu =0.15 ; Lung

PL = 0.50 ; Liver

PS = 0.30 ; Spleen

PK = 0.01 ; Kidneys

PT = 0.25 ; Tumor

PBo = 0.15 ; Rest of body

; Diffusion limitation coefficient constants, unitless

PALuC = 0.001 ; Lung

PALC = 0.001 ; Liver

PASC = 0.001 ; Spleen

PAKC = 0.0001 ; Kidneys

PATC = 0.001 ; Tumor

PABoC = 0.00001 ; Rest of body

; Permeability coefficient-surface area cross-product

PALu = PALuC*QLu ; Lung

PAL = PALC*QL ; Liver

PAS = PASC*QS ; Spleen

PAK = PAKC*QK ; Kidneys

PAT = PATC*QT ; Tumor

PABo = PABoC*QBo ; Rest of body

; Endocytosis-related parameters; Lu, L, K, T, and Bo represent the lung, liver, kidneys, tumor, and rest of body, respectively

KupLumax=0.075

KupLu50=24

KupLun=5

KoutLu=0.003

KupLu=((KupLumax*time^KupLun)/(KupLu50^KupLun+time^KupLun))

KupLmax=30

KupL50=48

KupLn=5

KoutL=0.001

KupL=((KupLmax*time^KupLn)/(KupL50^KupLn+time^KupLn))

KupSmax=60

KupS50=48

KupSn=5

KoutS=0.001

KupS=((KupSmax*time^KupSn)/(KupS50^KupSn+time^KupSn))

KupKmax=0.005

KupK50=48

KupKn=5

KoutK=0.0004

KupK=((KupKmax*time^KupKn)/(KupK50^KupKn+time^KupKn))

KupTmax = 20

KupT50=36

KupTn=5

KoutT=0.001

KupT=((KupTmax*time^KupTn)/(KupT50^KupTn+time^KupTn))

KupBomax=0.2

KupBo50=24

KupBon=5

KoutBo=0.0001

KupBo=((KupBomax*time^KupBon)/(KupBo50^KupBon+time^KupBon))

```
; Urine and biliary excretion
Kurine=0.00003
Kbile=0.00003

; Blood compartment
; Venous blood concentration
d/dt (AV)  = (QL*CVL+ QS*CVL + QK*CVK + QT*CVT + QBo*CVBo)-(QC*CV)
init AV = IV
CV = AV/VV

; Arterial blood concentration
d/dt (AA)  = QC*(CVLu-CA)
init AA = 0
CA = AA/VA

; Lung compartment; VES, T and RES represent blood vessels, tissue, and phagocytic cells
d/dt (ALuVES) = QLu*(CV-CVLu) - PALu*CVLu + (PALu*CLuT)/PLu
init ALuVES = 0
CVLu = ALuVES/VLuVES

d/dt (ALuT) = PALu*CVLu - (PALu*CLuT)/PLu + KoutLu *ALuRES - KupLu *ALuVES
init ALuT = 0
CLuT = ALuT/VLuT

d/dt (ALuRES) = KupLu *ALuVES - KoutLu *ALuRES
init ALuRES = 0
CLung = (ALuVES+ALuT+ALuRES)/VLu

; Liver compartment
d/dt (ALVES) = QL*CA+QS*CVS-(QL+QS)*CVL - PAL*CVL + (PAL*CLT)/PL + KoutL *ALRES - KupL
*ALVES - Kbile*CVL
init ALVES = 0
CVL = ALVES/VLVES
```

d/dt (Abile) = Kbile*CVL

init Abile = 0

d/dt (ALT) = PAL*CVL - (PAL*CLT)/PL

init ALT = 0

CLT = ALT/VLT

d/dt (ALRES) = KupL *ALVES - KoutL *ALRES

init ALRES = 0

CLiver = (ALVES+ALT+ALRES)/VL

; Spleen compartment

d/dt (ASVES) = QS*(CA-CVS) - PAS*CVS + (PAS*CST)/PS

init ASVES = 0

CVS = ASVES/VSVES

d/dt (AST) = PAS*CVS - (PAS*CST)/PS + KoutS *ASRES - KupS *ASVES

init AST = 0

CST = AST/VST

d/dt (ASRES) = KupS *ASVES - KoutS *ASRES

init ASRES = 0

CSpleen = (ASVES+AST+ASRES)/VS

; Kidney compartment

d/dt (AKVES) = QK*(CA-CVK) - PAK*CVK + (PAK*CKT)/PK - Kurine*CVK

init AKVES = 0

CVK = AKVES/VKVES

d/dt (Aurine) = Kurine*CVK

init Aurine = 0

```
d/dt (AKT) = PAK*CVK - (PAK*CKT)/PK + KoutK *AKRES - KupK *AKVES
init AKT = 0
CKT = AKT/VKT

d/dt (AKRES) = KupK *AKVES - KoutK *AKRES
init AKRES = 0
CKidney = (AKVES+AKT+AKRES)/VK

; Tumor compartment
d/dt (ATVES) = QT*(CA-CVT) - PAT*CVT + (PAT*CTT)/PT - Kurine*CVT
init ATVES = 0
CVT = ATVES/VTVES

d/dt (ATT) = PAT*CVT - (PAT*CTT)/PT + KoutT *ATRES - KupT *ATVES
init ATT = 0
CTT = ATT/VTT

d/dt (ATRES) = KupT *ATVES - KoutT *ATRES
init ATRES = 0
CTumor = (ATVES+ATT+ATRES)/VT

; Rest of body compartment
d/dt (ABoVES) = QBo*(CA-CVBo) - PABo*CVBo + (PABo*CBoT)/PBo
init ABoVES = 0
CVBo = ABoVES/VBoVES

d/dt (ABoT) = PABo*CVBo - (PABo*CBoT)/PBo + KoutBo *ABoRES - KupBo *ABoVES
init ABoT = 0
CBoT = ABoT/VBoT

d/dt (ABoRES) = KupBo *ABoVES - KoutBo *ABoRES
```

init ABoRES = 0

CBody = (ABoVES+ABoT+ABoRES)/Vbo

References

1. Brown RP, Delp MD, Lindstedt SL, Rhomberg LR, Beliles RP. Physiological Parameter Values for Physiologically Based Pharmacokinetic Models. Toxicol Ind Health. 1997; 13: 407-84.

2. Davies B, Morris T. Physiological Parameters in Laboratory Animals and Humans. Pharm Res. 1993; 10: 1093-5.

3. Baxter LT, Zhu H, Mackensen DG, Jain RK. Physiologically Based Pharmacokinetic Model for Specific and Nonspecific Monoclonal Antibodies and Fragments in Normal Tissues and Human Tumor Xenografts in Nude Mice. Cancer Res. 1994; 54: 1517-28.